Verknüpfte Excel-Tabellen in Access wieder aktualisieren, Teil 1
Versionen: Access 2003 und 2002/XP

Nach der Installation des Service Pack 2 für Office 2003 oder des Hotfixes KB904018 vom 18.10.2005 für Access 2002/XP lassen sich verknüpfte Excel-Tabellen nicht mehr aktualisiert werden. Beim Versuch, einen Feldinhalt zu ändern, ist lediglich ein akustischer Hinweis zu hören, eine Fehler- oder Statusmeldung wird nicht ausgegeben.

Viele Access-Anwender stehen damit vor ernsthaften Problemen, da unzählige Lösungen, die auf dieser Technologie basieren, von heute auf morgen nicht mehr funktionieren. In unserem zweiteiligen Tipp der Woche stellen wir Ihnen eine Technik vor, mit der sich das Problem zwar nicht komplett lösen, aber zumindest etwas kompensieren lässt.

Microsoft schlägt in einem Knowledge Base-Artikel zu diesem Thema vor, die betreffende Excel-Tabelle in Access zu importieren, die gewünschten Änderungen vorzunehmen und die Access-Tabelle dann wieder in eine Excel-Arbeitsmappe zu exportieren. Das mag in einigen wenigen Fällen praktikabel sein, aber wenn Sie Änderungen zum Beispiel täglich vornehmen müssen, ist dieser Weg zu umständlich. Wichtig wäre eine automatische Technik, um die Excel-Tabelle beim Öffnen der Datenbank zu importieren und die geänderten Daten beim Verlassen der Anwendung wieder in die Tabelle zu exportieren:

1. Löschen Sie zunächst die Verknüpfung mit der Excel-Tabelle im Bereich "Tabellen" des Datenbankfensters.

2. Legen Sie ein Modul "modExcelImpExp" an und geben Sie die folgenden Funktionen ein:

Function ImportExcelTab(strWorkbook As String, _
                        strExcelTabName As String, _
                        strAccTabName As String)

  On Error Resume Next
  ImportExcelTab = False 'Vorgabe: Problem
  DoCmd.SetWarnings False
  DoCmd.RunSQL "DROP TABLE " & strAccTabName
  DoCmd.SetWarnings True
  Err = 0
  DoCmd.TransferSpreadsheet acImport, _
        acSpreadsheetTypeExcel8, _
        strAccTabName, strWorkbook, True, _
        strExcelTabName & "!"
  If Err <> 0 Then
    Beep
    MsgBox "Fehler beim Importieren von '" & _
           strWorkbook & _
           "' (Tabelle: " & strExcelTabName & ")" & _
           vbCrLf & vbCrLf & Err.Description, _
           vbOKOnly + vbExclamation, "!!! Problem !!!"
  Else
    ImportExcelTab = True
  End If

End Function

Function AddSlash(aPath As String) As String

  AddSlash = aPath
  If Trim$(aPath) = "" Then Exit Function
  If Right$(aPath, 1) = "\" Then Exit Function
  AddSlash = aPath + "\"

End Function

Function GetDataBasePath() As String
  Dim db As DAO.Database, L As Integer, aPath As String

  Set db = CurrentDb()
  aPath = db.Name
  L = Len(aPath)
  While Mid$(aPath, L, 1) <> "\" And L > 0
      L = L - 1
  Wend
  If L > 1 Then
      aPath = Left$(aPath, L)
  Else
      aPath = ""
  End If
  GetDataBasePath = aPath

End Function

Function PathOnly(strPath As String) As String
  Dim L As Integer

  PathOnly = ""
  If InStr(strPath, "\") = 0 Then Exit Function
  L = Len(strPath)
  While Mid$(strPath, L, 1) <> "\" And L > 0
      L = L - 1
  Wend
  If L > 1 Then PathOnly = Left$(strPath, L)

End Function
· Setzen Sie über Extras-Verweise eine Referenz auf "Microsoft DAO Object Library 3.6" und auf "Microsoft Excel x.0 Object Library" (x.0= 11.0 für Excel 2003, 10.0 für Excel 2002/XP).

· Legen Sie ein "AutoExec"-Makro an oder ergänzen Sie ein vorhandenes "AutoExec"-Makro um die folgende Aktion:

Aktion: AusführenCode
Funktionsname: ImportExcelTab("<Arbeitsmappe>",
"<ExcelTabelle>", "<AccessTabelle>")

Für "Arbeitsmappe" geben Sie Pfad und Dateiname der Excel-Arbeitsmappe an, aus der eine Tabelle importiert werden soll. "ExcelTabelle" spezifiziert das zu importierende Tabellenblatt und "AccessTabelle" ist der Name der Tabelle, in die die Daten importiert werden. Diese Tabelle muss nicht vorhanden sein, vielmehr wird sie von der Funktion "ImportExcelTab()" bei Vorhandensein gelöscht.

· Speichern Sie das Makro und verlassen Sie den Makroentwurf.

Damit ist zunächst der Mechanismus aufgebaut, der eine Excel-Tabelle beim Öffnen der Datenbank importiert. Das geschieht künftig mit Hilfe der Funktion "ImportExcelTab()" wie folgt: Im ersten Schritt wird die Tabelle gelöscht, in die die Daten aus der Excel-Tabelle importiert werden sollen. Andernfalls würde die folgende Ausführung von "DoCmd.TransferSpreadsheet" scheitern, da es keine vorhandenen Tabellen ersetzen oder überschreiben kann. Das Löschen der Tabelle übernimmt eine SQL-Anweisung, die per "DoCmd.RunSQL" nicht nur schneller ausgeführt wird, sondern wesentlich einfacher ist, als das übliche Löschen über die "TableDefs"-Auflistung. Per "DoCmd.SetWarnings" schalten wir zuvor Sicherheitsabfragen von Access aus und anschließend wieder ein. Der zweite Schritt besteht dann darin, per "DoCmd.TransferSpreadsheet" aus der jeweiligen Arbeitsmappe (Parameter "strWorkbook" die gewünschte Excel-Tabelle (Parameter "strExcelTabName") in die Access-Tabelle (Parameter "strAccTabName") zu übertragen. Wichtig ist dabei, dass der Name der Excel-Tabelle um ein Ausrufezeichen ergänzt wird, damit Excel erkennt, dass ein komplettes Tabellenblatt adressiert wird. Kommt es beim Import zu einem Fehler, wird dieser entsprechend angezeigt.

Testen können Sie den Import bei Bedarf mit folgender Prozedur:

Sub TestImport()
  Dim strPath As String
  Dim strFName As String
  Dim strTabRange As String
  Dim fOK As Boolean

  strPath = GetDataBasePath()
  strFName = AddSlash(strPath) & "Testmappe.xls"
  strTabRange = "Tabelle1"
  fOK = ImportExcelTab(strFName, _
                       strTabRange, _
                       "ExcelImport")

End Sub


Im zweiten Teil dieses Tipps in der nächsten Ausgabe geht es dann um den automatischen Export der Excel-Tabelle beim Schließen der Datenbank.
Verknüpfte Excel-Tabellen in Access wieder aktualisieren, Teil 2
Versionen: Access 2003 und 2002/XP

Wie bereits berichtet lassen sich nach der Installation des Service Pack 2 für Office 2003 oder des Hotfixes KB904018 vom 18.10.2005 für Access 2002/XP verknüpfte Excel-Tabellen nicht mehr aktualisieren. In unserem zweiteiligen Tipp der Woche stellen wir Ihnen eine Technik vor, mit der sich das Problem zwar nicht komplett lösen, aber zumindest etwas kompensieren lässt.

Im ersten Teil dieses Tipps der Woche ging es dabei um den automatisierten Import einer Excel-Tabelle beim Öffnen einer Datenbank. Nun muss die Excel-Tabelle noch automatisch beim Schließen der Datenbank exportiert werden.

Access stellt mit dem "AutoExec"-Makro zwar eine Möglichkeit bereit, Aktionen beim Öffnen einer Datenbank automatisch auszuführen, aber ein "AutoClose"-Makro bzw. einen vergleichbaren Mechanismus, der beim Schließen Aktionen automatisch ausführt, gibt nicht. Für den Export der geänderten Excel-Daten müssen wir also zu einem Trick greifen: Wir setzen ein Formular ein, das beim Öffnen der Datenbank automatisch versteckt über "AutoExec" geladen wird. Beim Verlassen der Datenbank schließt Access der Reihe nach alle offenen Formulare und ruft dabei die Ereignisprozedur "Beim Schließen" des jeweiligen Formulars und somit auch die Ereignisprozedur des unsichtbaren Formulars auf. Hier werden dann die für den Export notwendigen Anweisungen ausgeführt:

· Legen Sie in der Datenbank zunächst ein neues, leeres Formular an.

· Wählen Sie das Menü Ansicht-Eigenschaften an.

· Setzen Sie die Eigenschaft "Beim Schließen" auf den Eintrag "[Ereignisprozedur]" und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

· Im VBA-Editor geben Sie nun die Anweisungen ein, die beim Schließen der Datenbank ausgeführt werden sollen:

Private Sub Form_Close()
  Dim strPath As String
  Dim strFName As String
  Dim strExcelTabName As String
  Dim strAccTabName As String
  Dim fOK As Boolean

  strPath = GetDataBasePath()
  strFName = AddSlash(strPath) & "Testmappe.xls"
  strExcelTabName = "Tabelle1"
  strAccTabName = "ExcelImport"
  fOK = ExportExcelTab(strFName, _
                       strTabRange, _
                       strAccTabName)

End Sub
· Passen Sie die Zuweisungen zu den Variablen "strPath", strFName", "strExcelTabName" und "strAccTabName" entsprechend an und speichern Sie das Formular unter dem Namen "frmAutoClose".

· Fügen Sie im Makro "AutoExec" am Anfang eine Aktion "ÖffnenFormular" hinzu, wählen Sie im Feld "Formularname" den Eintrag "frmAutoClose" und im Feld "Fenstermodus" den Eintrag "Ausgeblendet".

· Sichern Sie die Änderungen am Makro und verlassen Sie den Makroentwurf.

Wir gehen hier davon aus, dass eine Excel-Arbeitsmappe "Testmappe.xls" im gleichen Verzeichnis wie die Datenbank gespeichert ist. Die Funktion "GetDatabasePath()" (siehe Teil 1 des Tipps der Woche) liefert den Pfad der Datenbank (strPath), den wir dann um den Dateinamen der Excel-Arbeitsmappe ergänzen (strFName). Die zu exportierenden Daten sollen in dieser Arbeitsmappe im Tabellenblatt "Tabelle1" abgelegt werden (strExcelTabName), der Ursprung der Daten ist die Access-Tabelle "ExcelImport" (strAccTabNnme). Diese Parameter übergeben wir dann an die Funktion "ExportExcelTab()", die Sie dazu im Modul "modExcelImpExp", das in Teil 1 des Tipps der Woche angelegt wurde, eingeben:

Function ExportExcelTab(strWorkbook As String, _
                        strExcelTabName As String, _
                        strAccTabName As String)

  Dim objExcel As Excel.Application
  Dim wb As Excel.Workbook
  Dim ws As Excel.Worksheet
  Dim strNewName As String

  ExportExcelTab = False 'Vorgabe: Problem
  On Error GoTo ExcelError
  Set objExcel = CreateObject("Excel.Application")
  Set wb = objExcel.Workbooks.Open(strWorkbook)
  On Error Resume Next
  Err = 0
  objExcel.Visible = True
  Set ws = wb.Sheets(strExcelTabName)
  If Err = 0 Then
    'Tabellenblatt vorhanden, umbenennen mit DZ-Stempel
    strNewName = ws.Name & " " & _
                 Format$(Now, "yyyy-mm-dd hh-nn")
    ws.Name = strNewName
    ws.Move After:=wb.Sheets(wb.Sheets.Count)
    Set ws = wb.Sheets.Add
    ws.Name = strExcelTabName
    ws.Move Before:=wb.Sheets(1)
    ws.Range("A1").Select
  End If
  wb.Close True

  Err = 0
  DoCmd.TransferSpreadsheet acExport, _
        acSpreadsheetTypeExcel8, _
        strAccTabName, strWorkbook, True, _
        strExcelTabName & "!"
  If Err <> 0 Then
    Beep
    MsgBox "Fehler beim Exportieren von Tabelle '" & _
           strAccTabName & _
           "' in '" & strWorkbook & "'" & _
           " (Tabelle: " & strExcelTabName & ")" & _
           vbCrLf & vbCrLf & Err.Description, _
           vbOKOnly + vbExclamation, "!!! Problem !!!"
  Else
    ExportExcelTab = True
  End If

ExitFunc:
  On Error Resume Next
  objExcel.Quit
  Set ws = Nothing
  Set wb = Nothing
  Set objExcel = Nothing
  DoEvents
  Err.Clear
  Exit Function

ExcelError:
  Beep
  MsgBox "Fehler beim Zugriff auf Excel: " & _
         Err.Description, _
         vbOKOnly + vbExclamation, "!!! Problem !!!"
  Resume ExitFunc

End Function

Um sicherzustellen, dass Daten in der Excel-Arbeitsmappe nicht unwiederbringlich überschrieben werden, legen wir vor dem Export grundsätzlich eine Sicherungskopie des betreffenden Tabellenblattes unter dem ursprünglichen Namen an; ergänzt um einen Datums-/Zeitstempel. Dazu wird die Excel-Arbeitsmappe per Office Automation geöffnet, das Tabellenblatt über die "Sheets"-Auflistung adressiert und wie erläutert umbenannt. Danach verschieben wir es mit der Methode "Move" ans Ende der Tabellenblätter. Anschließend legen wir ein neues Tabellenblatt an, benennen es wie das Tabellenblatt, in das die Daten geschrieben werden sollen, verschieben es an den Anfang der Tabellenblätter und aktivieren die erste Zelle. Dann schließen wir die Arbeitsmappe wieder, damit es keine Konflikte mit dem gleich folgenden "DoCmd.TransferSpreadsheet" gibt. Der Parameter "True" sorgt dabei dafür, dass die Arbeitsmappe ohne Rückfrage "Änderungen speichern?" gesichert wird.

Der zweite Schritt besteht darin, per "DoCmd.TransferSpreadsheet" die Access-Tabelle (Parameter "strAccTabName") in die jeweilige Arbeitsmappe (Parameter "strWorkbook" und dort in die gewünschte Excel-Tabelle (Parameter "strExcelTabName") zu übertragen. Wichtig ist auch hier wieder, dass der Name der Excel-Tabelle um ein Ausrufezeichen ergänzt wird, damit Excel erkennt, dass ein komplettes Tabellenblatt adressiert wird. Kommt es beim Export zu einem Fehler, wird er angezeigt. Den Abschluss bilden dann nur noch einige "Aufräumarbeiten". Für die Fehlerbehandlung schalten wir übrigens zwischen individueller Behandlung (On Error Resume Next, If Err<> 0 Then...) und einer zentralen Fehlerbehandlung für Excel-Zugriffe (ExcelError) um.

Den Export können Sie bei Bedarf mit der folgenden Prozedur testen:

Sub TestExport()
  Dim strPath As String
  Dim strFName As String
  Dim strExcelTabName As String
  Dim strAccTabName As String
  Dim fOK As Boolean

  strPath = GetDataBasePath()
  strFName = AddSlash(strPath) & "Testmappe.xls"
  strExcelTabName = "Tabelle1"
  strAccTabName = "ExcelImport"
  fOK = ExportExcelTab(strFName, strExcelTabName, _
  strAccTabName)

End Sub

